# $Id: PKGBUILD 89809 2013-05-02 13:39:19Z arodseth $
# Maintainer: Alexander Rødseth <rodseth@gmail.com>
# Contributor: Jaroslav Lichtblau <dragonlord@aur.archlinux.org>
# Contributor: Allan McRae <allan@archlinux.org>
# Contributor: David Moore <davidm@sjsoft.com>

pkgbase=python-simplejson
pkgname=("$pkgbase" 'python2-simplejson')
pkgver=3.2.0
pkgrel=1
pkgdesc='Simple, fast, extensible JSON encoder/decoder for Python'
license=('MIT')
arch=('x86_64' 'i686')
url='http://undefined.org/python/#simplejson'
makedepends=('python' 'python2-distribute')
source=("$pkgname-$pkgver.tar.gz::https://github.com/simplejson/simplejson/tarball/v${pkgver}")
sha256sums=('507036d10039ccf437625437b2521c0f691c20c3911b2b7183d5a66c98d36432')

package_python-simplejson() {
  depends=('python')

  cd "$srcdir/simplejson-simplejson-"*
  python setup.py install --root="$pkgdir"
  install -Dm644 "$srcdir/simplejson-simplejson"-*/LICENSE.txt \
    "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

package_python2-simplejson() {
  depends=('python2')

  cd "$srcdir/simplejson-simplejson-"*
  python2 setup.py install --root="$pkgdir"
  install -Dm644 "$srcdir/simplejson-simplejson"-*/LICENSE.txt \
    "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

# vim:set ts=2 sw=2 et:
